Lesson #6: Probability Distributions.

Some Introductory Demonstrations

- **Connecting the CDF and the PDF**
- **Mean and Standard Deviation of a Distribution**

A Warm Up

In[22]:=
```
f[x_] := λ E^(-λ x)

Integrate[f[x], {x, 0, Infinity}]
```

$\lambda\,\text{If}\left[\text{Re}[\lambda] > 0,\ \frac{1}{\lambda},\ \text{Integrate}\left[e^{-x\lambda},\ \{x, 0, \infty\},\ \text{Assumptions} \to \text{Re}[\lambda] \le 0\right]\right]$

```
Integrate[f[x], {x, 0, Infinity}, Assumptions → Re[λ] > 0]
```

1

```
Integrate[f[x] x, {x, 0, Infinity}, Assumptions → Re[λ] > 0]
```

$\frac{1}{\lambda}$

```
Plot[Table[f[x] /. λ → lam, {lam, {1, 5, 10}}], {x, 0, 2}, PlotRange → All]
```



```
Manipulate[
 Plot[f[x] /. λ → lam, {x, 0, 2}, PlotRange → {0, 10 },
  AxesOrigin → {0, 0}, Epilog → {Thick, Red, Line[{{1/lam, 0}, {1/lam, 10}}]}]
 , {lam, 1/2, 10}]
```



Mathematica' s Built In Probability Distributions

```
? *Distribution*
```

▼ System`

BernoulliDistribution	InverseGaussianDistribution
BetaBinomialDistribution	LaplaceDistribution
BetaDistribution	LogisticDistribution
BetaNegativeBinomialDistribution	LogNormalDistribution
BinomialDistribution	LogSeriesDistribution
CauchyDistribution	MaxwellDistribution
ChiDistribution	NegativeBinomialDistribution
ChiSquareDistribution	NoncentralChiSquareDistribution
DiscreteUniformDistribution	NoncentralFRatioDistribution
DistributionDomain	NoncentralStudentTDistribution
DistributionDomainQ	NormalDistribution
DistributionParameterQ	ParetoDistribution
ExponentialDistribution	PoissonDistribution
ExtremeValueDistribution	RayleighDistribution
FRatioDistribution	StudentTDistribution
GammaDistribution	TriangularDistribution
GeometricDistribution	UniformDistribution
GumbelDistribution	WeibullDistribution
HalfNormalDistribution	ZipfDistribution
HypergeometricDistribution	

```
? PDF
```

PDF[*dist*, *x*] gives the probability density function for the symbolic distribution *dist* evaluated at *x*.
PDF[*dist*] gives the PDF as a pure function. ≫

```
PDF[NormalDistribution[μ, σ], x]
```

$$\frac{e^{-\frac{(x-\mu)^2}{2\sigma^2}}}{\sqrt{2\pi}\,\sigma}$$

```
Plot[% /. {μ → 0, σ → 1}, {x, -3, 3}]
```



```
?CDF
```

CDF[*dist*, *x*] gives the cumulative distribution function for the symbolic distribution *dist* evaluated at *x*.
CDF[*dist*] gives the CDF as a pure function. ≫

```
CDF[NormalDistribution[μ, σ], x]
```

$$\frac{1}{2}\left(1+\text{Erf}\left[\frac{x-\mu}{\sqrt{2}\,\sigma}\right]\right)$$

```
CDF[ExponentialDistribution[λ], x]
```

$$\begin{cases} 1-e^{-x\lambda} & x>0 \end{cases}$$

```
Integrate[1 - %, {x, 0, Infinity}, Assumptions → λ > 0]
```

$$\frac{1}{\lambda}$$

```
?CharacteristicFunction
```

CharacteristicFunction[*dist*, *t*] gives the characteristic
function for the symbolic distribution *dist* as a function of the variable *t*. ≫

```
CharacteristicFunction[ExponentialDistribution[λ], s]
```

$$\frac{\lambda}{-i\, s+\lambda}$$

```
CharacteristicFunction[ExponentialDistribution[λ], -I s]
```

$$\frac{\lambda}{-s+\lambda}$$

```
D[%, s] /. s → 0
```

$$\frac{1}{\lambda}$$

```
Table[D[CharacteristicFunction[ExponentialDistribution[λ], -I s], {s, n}] /. s → 0, {n, 15}]
```

$$\left\{\frac{1}{\lambda},\frac{2}{\lambda^2},\frac{6}{\lambda^3},\frac{24}{\lambda^4},\frac{120}{\lambda^5},\frac{720}{\lambda^6},\frac{5040}{\lambda^7},\frac{40\,320}{\lambda^8},\frac{362\,880}{\lambda^9},\frac{3\,628\,800}{\lambda^{10}},\frac{39\,916\,800}{\lambda^{11}},\frac{479\,001\,600}{\lambda^{12}},\frac{6\,227\,020\,800}{\lambda^{13}},\frac{87\,178\,291\,200}{\lambda^{14}},\frac{1\,307\,674\,368\,000}{\lambda^{15}}\right\}$$

```
CharacteristicFunction[NormalDistribution[μ, σ], s]
```

$$e^{i\, s\,\mu-\frac{s^2\sigma^2}{2}}$$

$$\frac{\texttt{D[\%, s] /. s → 0}}{\texttt{I}}$$

$$\mu$$

$$\frac{\texttt{D[\%55, \{s, 2\}] /. s → 0}}{\texttt{-1}}$$

$$\mu^2+\sigma^2$$

```
% - μ^2
```

σ^2

```
? Mean
```

Mean[*list*] gives the statistical mean of the elements in *list*.
Mean[*dist*] gives the mean of the symbolic distribution *dist*. ≫

```
Mean[NormalDistribution[μ, 12]]
```

μ

```
Mean[{1, 2, 3.01}]
```

2.00333

```
? ExpectedValue
```

ExpectedValue[*f*, *dist*] gives the expected value of the pure function *f* with respect to the symbolic distribution *dist*.
ExpectedValue[*f*, *dist*, *x*] gives the expected value of the function *f* of *x* with respect to the symbolic distribution *dist*. ≫

```
ExpectedValue[(# - Mean[ExponentialDistribution[λ]])^2 &, ExponentialDistribution[λ]]
```

$$\frac{1}{\lambda^2}$$

```
ExpectedValue[If[x >= 10, 1, 0], ExponentialDistribution[λ], x]
```

$e^{-10 \lambda}$

```
? Variance
```

Variance[*list*] gives the statistical variance of the elements in *list*.
Variance[*dist*] gives the variance of the symbolic distribution *dist*. ≫

```
? CentralMoment
```

CentralMoment[*list*, *r*] gives the r^{th} central moment of the elements in *list* with respect to their mean. ≫

```
? RandomReal
```

RandomReal[] gives a pseudorandom real number in the range 0 to 1.
RandomReal[{x_{min}, x_{max}}] gives a pseudorandom real number in the range x_{min} to x_{max}.
RandomReal[x_{max}] gives a pseudorandom real number in the range 0 to x_{max}.
RandomReal[*range*, *n*] gives a list of *n* pseudorandom reals.
RandomReal[*range*, {n_1, n_2, …}] gives an $n_1 \times n_2 \times \ldots$ array of pseudorandom reals.
RandomReal[*dist*, …] samples from the symbolic continuous distribution *dist*. ≫

```
dat = Table[RandomReal[ExponentialDistribution[2]], {1000}];
Variance[dat] - Variance[ExponentialDistribution[2]]
```

0.001277

■ There are also Discrete Distributions (PDF means "mass" in this case)

```
PDF[BinomialDistribution[n, p], k]
```

$(1-p)^{-k+n}\, p^k\, \mathrm{Binomial}[n, k]$

```
?Binomial
```

Binomial[*n*, *m*] gives the binomial coefficient $\binom{n}{m}$. ≫

```
ListPlot[
 Table[{k, PDF[BinomialDistribution[10, p], k]} /. p → 1/2, {k, 0, 10}], Filling → Axis]
```




```
Manipulate[
 ListPlot[Table[{k, PDF[BinomialDistribution[10, p], k]} /. p → pp, {k, 0, 10}],
  PlotRange → {0, 0.3}, Filling → Axis,
  PlotStyle → {PointSize[0.05]}, FillingStyle → Thickness[0.03]]
 , {pp, 0.3, 0.7}]
```



Now Lets Learn Some New Stuff (maybe review):

The Log-Normal Distribution.

Concepts to Understand:

- Transformations of Probability Distributions (in this case Log and Exponent).
- Usage of Log-Normal for Multiplicative Data.

The Cauchy Distribution

Concepts to Understand:

- "Physical" creation of the Cauchy Distribution.
- A distribution without an expectation.
- Transformation of two probability distirubtions (in this case a quetient of two Normals).

Hazarad Rates

Concepts to Understand

- What is the Hazarad Rate.
- Examples.

Class Exercise: The Gamma Distribution

The Gamma Density has the "basic form" :

```
Plot[
 {E^-x,      x,        E^-x x}
 , {x, 0, 3}, PlotStyle → {Dashed, Dashed, {Black, Thick}}]
```



We can parametrize it :

```
Manipulate[
 Plot[{E^(-λ x), x^(α-1), E^(-x λ) x^(α-1)}, {x, 0, 5},
  PlotStyle → {Dashed, Dashed, {Black, Thick}}, PlotRange → {0, 5}],
 {{λ, 1}, 0.5, 3}, {{α, 2}, 0.5, 3}]
```



Need to "normalize" it (integral on suppurt needs to be 1) :

```
Integrate[E^(-x λ) x^(α-1), {x, 0, Infinity}, Assumptions → {α > 0, λ > 0}]
```

```
λ^-α Gamma[α]
```

```
? Gamma
```

Gamma[*z*] is the Euler gamma function Γ(z).
Gamma[*a*, *z*] is the plica function Γ(a, z).
Gamma[*a*, z_0, z_1] is the generalized plica function $\Gamma(a, z_0) - \Gamma(a, z_1)$. ≫

```
Integrate[1/(λ^-α Gamma[α]) E^(-x λ) x^(α-1), {x, 0, Infinity}, Assumptions → {α > 0, λ > 0}]
```

```
1
```

```
ourPDF[x_] := λ^α/Gamma[α] E^(-x λ) x^(α-1)
```

Now lets look at the built in Gamma Distirubtion

```
? GammaDistribution
```

GammaDistribution[*α*, *β*] represents a gamma distribution with shape parameter *α* and scale parameter *β*. ≫

```
PDF[GammaDistribution[α, β], x]
```

$$\frac{e^{-\frac{x}{\beta}} x^{-1+\alpha} \beta^{-\alpha}}{\text{Gamma}[\alpha]}$$

```
Mean[GammaDistribution[α, β]]
```

$\alpha\,\beta$

```
Integrate[ourPDF[x] x, {x, 0, Infinity}, Assumptions → {α > 0, λ > 0}]
```

$\frac{\alpha}{\lambda}$

- ### Exercise 1 : Exponential Distribution as a special case
- ### Exercise 2 : χ^2 Distribution as a special case
- ### Exercise 3 : Sums of Gamma random variables